

□ **Modeling Change: Springs, Driving Safety, Radioactivity, Trees, Fish, and Mammals**

☞ (Chapter 1 - Technology Application Projects)

□ **1 Introduction**

☞ OBJECTIVE: Practice a modeling process: consider a behavior, observe data, fit a model, analyze the error, improve the model if appropriate, interpret the model, and make predictions.

Mathematical models help us better understand things in the world around us, like the behavior of springs, safe driving practices, radioactivity in medicine, the growth of trees and fish, and the biology of mammals. In this module you will learn how to construct mathematical models, how analyze and improve them, and how to use them to study the behavior you are modeling, to make predictions. As you will see, a computer algebra system like Maxima is a powerful tool that will aid you in your mathematical modeling.

In this worksheet, we use the two Maxima packages draw and stats. We initialize them at the beginning Part I.

□ **1.1 Technology Guidelines**

NOTE: If you have just finished a document, restart Maxima before executing a new document. This can be done by choosing "Restart" from the Maxima menu.

TO OPEN OR CLOSE CELLS

Click on the arrow at the top of the cell bracket.

TO STOP AN EXECUTION

Click on STOP button from the toolbar.

ORDER OF EXECUTION

Execute commands in the order given. Do not skip any Maxima Input lines within a given document.

Alternatively, you can execute the entire worksheet by selecting the "Evaluate All Cells" command from the "Cell" drop down menu or simply press Ctrl-r.

SAVING WORKSHEETS

You can save anytime to any directory you choose, and it is wise to save often.

EXPERIENCING MAJOR PROBLEMS

Save if appropriate, and then shut down Maxima and start it up again.

2 Part I: Spring Elongation

2.1 Observing the Data

The response of a spring to various loads can be modeled in order to design a vehicle such as a tank, utility vehicle, or luxury car that responds to road conditions in a desired way. (See also "Bungee Cord Jumping: A Classroom Experiment," another module in this supplement.) We conducted an experiment to measure the stretch of a spring in inches as a function of the number of units of mass placed on the spring. The following list includes these data with the first element in each ordered pair being the mass on the spring and the second element its corresponding stretch.

```
(%i1) load(draw)$
      load(stats)$
      load(lsqares)$
      ratprint: false$
```

```
;Compiler warnings for "C:/PROGRA~1/MAXIMA~2.0/share/maxima/5.25.0/share/lbfgs/lb1.lisp" :
; In LB1: Unused lexical variable l
```

```
(%i5) data: [[0, 0], [1, .875], [2, 1.721], [3, 2.641], [4, 3.531], [5, 4.391],[6, 5.241], [7, 6.120],  
            [8, 6.992], [9, 7.869], [10, 8.741]];  
(%o5) [[0,0],[1,0.875],[2,1.721],[3,2.641],[4,3.531],[5,4.391],[6,5.241],[7,6.12],[8,  
6.992],[9,7.869],[10,8.741]]
```


```
(%i6) datamatrix: matrix(["mass","stretch (in)],[0, 0], [1, .875], [2, 1.721], [3, 2.641], [4, 3.531],  
                        [5, 4.391],[6, 5.241], [7, 6.120], [8, 6.992], [9, 7.869], [10, 8.741]]);  
(%o6) 

| <i>mass</i> | <i>stretch (in)</i> |
|-------------|---------------------|
| 0           | 0                   |
| 1           | 0.875               |
| 2           | 1.721               |
| 3           | 2.641               |
| 4           | 3.531               |
| 5           | 4.391               |
| 6           | 5.241               |
| 7           | 6.12                |
| 8           | 6.992               |
| 9           | 7.869               |
| 10          | 8.741               |


```

Next, we plot the data to see if there is a recognizable pattern and name the plot p1 for later use.

```
(%i7) p1: wxdraw2d(  
    nticks=200,  
    color=blue,  
    point_type=6,  
    point_size=1,  
    points(data),  
    xlabel="mass",  
    ylabel="stretch (in)");
```



(%o7)

□ 2.2 Designing a Model

The data strongly suggest a linear relationship. Now we use the `simple_linear_regression()` function to find the line of best-fit or the linear regression line.

The output of the function `simple_linear_regression` is an `inference_result` Maxima object with the following results:

1. `model`: the equation for the line of best fit.
2. `means`: bivariate mean.
3. `variances`: variances of both variables.
4. `correlation`: correlation coefficient.
5. `adc`: adjusted determination coefficient.
6. `a_estimation`: estimation of parameter a.
7. `a_conf_int`: confidence interval of parameter a.
8. `b_estimation`: estimation of parameter b.
9. `b_conf_int`: confidence interval of parameter b.
10. `hypotheses`: null and alternative hypotheses about parameter b.
11. `statistic`: value of the sample statistic used for testing the null hypothesis.
12. `distribution`: distribution of the sample statistic, together with its parameter.
13. `p_value`: p-value of the test about b.
14. `v_estimation`: unbiased variance estimation, or residual variance.
15. `v_conf_int`: variance confidence interval.
16. `cond_mean_conf_int`: confidence interval for the conditioned mean.
17. `new_pred_conf_int`: confidence interval for a new prediction.
18. `residuals`: list of pairs (prediction, residual), ordered with respect to predictions. This is useful for goodness of fit analysis.

Only items 1, 4, 14, 9, 10, 11, 12, and 13 above, in this order, are shown by default. The rest remain hidden until the user makes use of functions `items_inference` and `take_inference`.

```
(%i8) data2: simple_linear_regression(data);
```

```
      SIMPLE LINEAR REGRESSION
```

```
      model = 0.8742 x + 0.0037272727272727413
```

```
      correlation = 0.999984953725526
```

```
      v_estimation = 2.810868686868712 10-4
```

```
(%o8) b_conf_int = [0.870583847725235, 0.877816152274765]
```

```
      hypotheses = H0: b = 0, H1: b ≠ 0
```

```
      statistic = 546.8734834174389
```


```
      distribution = [student_t, 9]
```

```
      p_value = 0.0
```


```
(%i9) eqn: take_inference(model,data2);  
(%o9) 0.8742 x+0.00372727272727413
```

The constant of proportionality is 0.8742. Next, we plot the regression line, save it as p2, and then show the plots of the data and the regression line together on the same graph.

```
(%i10) p2: wxdraw2d(  
    nticks=200,  
    color=red,  
    explicit(eqn,x,0,10),  
    xlabel="mass",  
    ylabel="stretch (in)");
```



```
(%i11) wxdraw2d(  
    nticks=200,  
    color=blue,  
    point_type=6,  
    point_size=1,  
    points(data),  
    color=red,  
    explicit(eqn,x,0,10),  
    xlabel="mass",  
    ylabel="stretch (in)");
```



(%o11)

2.3 Assessing the Errors

As we expected, the line appears to fit the data very well. We can provide further verification by calculating the residual errors for each data point and plotting them. First, we need to use the model to calculate the values of stretch for each mass value in the original data set.

The residual error (or residual) is the difference between the measured value and the value the model predicts for each amount of mass.

Note that a list of predicted value / residual pairs is automatically created when the `simple_linear_regression()` function is executed. So we will retrieve this list, and then save just the residual portion as a new list. Next we create the residual plot.

```
(%i12) kill(residuals);
      PredictValues_Residuals: take_inference(residuals,data2);
```

```
(%o12) done
```


```
(%o13) [[0.00372727272727413, -0.00372727272727413], [0.877927272727274, -
0.00292727272727389], [1.752127272727274, -0.0311272727272736], [2.626327272727274,
0.0146727272727265], [3.500527272727273, 0.030472727272727], [4.374727272727273,
0.0162727272727272], [5.248927272727273, -0.00792727272727323], [6.123127272727272, -
0.00312727272727198], [6.997327272727272, -0.00532727272727218], [7.871527272727272, -
0.00252727272727249], [8.745727272727272, -0.00472727272727269]]
```

```
(%i14) residuals: makelist([i,PredictValues_Residuals[i+1][2]],i,0,10);
```

```
(%o14) [[0, -0.00372727272727413], [1, -0.00292727272727389], [2, -0.0311272727272736], [3,
0.0146727272727265], [4, 0.030472727272727], [5, 0.0162727272727272], [6, -
0.00792727272727323], [7, -0.00312727272727198], [8, -0.00532727272727218], [9, -
0.00252727272727249], [10, -0.00472727272727269]]
```

```
(%i15) wxdraw2d(
      yrange=[-.04,.04],
      xrange=[-1,11],
      xlabel="mass",
      ylabel="residual error",
      color=blue,
      point_type=6,
      point_size=1,
      points(residuals));
```

```
(%t15)
```



```
(%o15)
```

The residuals show that the differences are indeed small. Small is a relative term, and sometimes it is more meaningful to look at the error in relation to the size of the quantity being estimated. For all but the first data point, we calculate the relative error by dividing the measured value into the residual and multiplying by 100 to express this relative error as a percentage. Then, we plot the percentage error for each data point.


```
(%i16) percenterrors: makelist([i,100*residuals[i+1][2]/data[i+1][2]],i,1,10);
```

```
(%o16) [[1,-0.334545454545587],[2,-1.808673604141404],[3,0.555574679012742],[4,0.863005586879835],[5,0.37059274135111],[6,-0.151254965221775],[7,-0.0510992275698036],[8,-0.0761909714998882],[9,-0.0321168220520077],[10,-0.0540816008153837]]
```

```
(%i17) percenterrplot: points(percenterrors);
```

```
(%o17) points([[1,-0.334545454545587],[2,-1.808673604141404],[3,0.555574679012742],[4,0.863005586879835],[5,0.37059274135111],[6,-0.151254965221775],[7,-0.0510992275698036],[8,-0.0761909714998882],[9,-0.0321168220520077],[10,-0.0540816008153837]])
```

```
(%i18) wxdraw2d(
  yrange=[-2,1],
  xrange=[-2,12],
  xlabel="mass",
  ylabel="% error",
  color=blue, percenterrplot);
```



(%o18)

3 Part II: Safe Following Distance

□ 3.1 Observing the Data

A rule of thumb that is often given for safe following distance is to allow 2 seconds between your car and the car in front of you. Is this rule reasonable?

The following data set contains ordered pairs of values. The first element of each ordered pair is the traveling speed, and the second element is the total stopping distance, the distance traveled by the car during the driver's reaction time plus the distance required for the vehicle to come to a stop with full braking.

```
(%i19) kill(all)$
      load(draw)$
      load(stats)$
      load(lsqares)$
      ratprint: false$
```

```
(%i5) data: [[20., 22 + 32], [25, 28 + 47], [30, 33 + 65], [35, 39 + 87], [40, 44 + 112],
             [45, 50 + 140], [50, 55 + 171], [55, 61 + 204], [60, 66 + 241], [65, 72 + 282],
             [70, 77 + 325], [75, 83 + 376]];

(%o5) [[20, 54], [25, 75], [30, 98], [35, 126], [40, 156], [45, 190], [50, 226], [55, 265], [60, 307],
       [65, 354], [70, 402], [75, 459]]
```

```
(%i6) datalabelsmatrix: matrix(["speed", "stopping dist"], [20., 22 + 32], [25, 28 + 47],
                               [30, 33 + 65], [35, 39 + 87], [40, 44 + 112], [45, 50 + 140], [50, 55 + 171],
                               [55, 61 + 204], [60, 66 + 241], [65, 72 + 282], [70, 77 + 325], [75, 83 + 376]);
```


```
(%o6)
      speed  stopping dist
      20      54
      25      75
      30      98
      35     126
      40     156
      45     190
      50     226
      55     265
      60     307
      65     354
      70     402
      75     459
```

```
(%i7) datamatrix: matrix([20., 22 + 32], [25, 28 + 47], [30, 33 + 65], [35, 39 + 87],  
[40, 44 + 112], [45, 50 + 140], [50, 55 + 171], [55, 61 + 204], [60, 66 + 241],  
[65, 72 + 282], [70, 77 + 325], [75, 83 + 376]);
```

```
(%o7) [ 20  54  
      25  75  
      30  98  
      35 126  
      40 156  
      45 190  
      50 226  
      55 265  
      60 307  
      65 354  
      70 402  
      75 459]
```

☑ We plot the data to see if there is a recognizable pattern.

```
(%i8) p1: wxdraw2d(  
  xlabel="speed (mph)",  
  ylabel="stopping distance (ft)",  
  color=blue,  
  point_type=6,  
  point_size=1,  
  points(data));
```



□ 3.2 Designing a Model

∟ The data suggest a possible quadratic relationship. Now we load the lsquares package in order to run a nonlinear regression. We will use the lsquares_estimates function to find the curve of best-fit.

```
∟ (%i9) load(lsquares)$  
      numer:true$  
      ratprint: false$
```

```
∟ (%i12) coeff: lsquares_estimates(datamatrix, [x,y], y=a+b*x+c*x^2, [a,b,c]);  
∟ (%o12) [[a=5.03996003996004, b=1.18051948051948, c=0.0645554445554446]]
```

```
∟ (%i13) a: rhs(coeff[1][1]);  
∟ (%o13) 5.03996003996004
```

```
∟ (%i14) b: rhs(coeff[1][2]);  
∟ (%o14) 1.18051948051948
```


```
∟ (%i15) c: rhs(coeff[1][3]);  
∟ (%o15) 0.0645554445554446
```

```
∟ (%i16) eqn: a+b*x+c*x^2;  
∟ (%o16) 0.0645554445554446 x2 + 1.18051948051948 x + 5.03996003996004
```

∟ Next, we plot the regression curve and then show the plot of the data and the model together on the same graph.


```
(%i17) p2: wxdraw2d(  
    nticks=10,  
    color=red,  
    explicit(eqn,x,0,85),  
    xlabel="speed (mph)",  
    ylabel="stopping distance (ft)");
```

(%t17)



(%o17)

```
(%i18) wxdraw2d(
    nticks=10,
    color=red,
    explicit(eqn,x,0,85),
    color=blue,
    point_type=6,
    point_size=1,
    points(data),
    xlabel="speed (mph)",
    ylabel="stopping distance (ft)");
```



The best-fit quadratic function appears to fit the data very well.

3.3 Assessing the Errors

We can provide further verification by calculating the residual errors for each data point and plotting them. First, we calculate the stopping distances predicted by the model, and then we calculate the residual errors and plot them.

```
(%i19) predictvalues: makelist([i*5,subst(i*5,x,eqn)],i,4,15);
```

```
(%o19) [[20,54.47252747252747],[25,74.90009990009999],[30,98.55544455544455],[35,125.4385614385614],[40,155.5494505494505],[45,188.8881118881119],[50,225.4545454545454],[55,265.2487512487512],[60,308.2707292707292],[65,354.5204795204795],[70,403.998001998002],[75,456.7032967032967]]
```

```
(%i20) predictvalueslabels: matrix(["speed","predicted values"],
    [20,54.47252747252747],[25,74.9000999000999],[30,98.55544455544455],
    [35,125.4385614385614],[40,155.5494505494505],[45,188.8881118881119],
    [50,225.4545454545454],[55,265.2487512487512],[60,308.2707292707293],
    [65,354.5204795204795],[70,403.998001998002],[75,456.7032967032967]);
```

```
(%o20)


| <i>speed</i> | <i>predicted values</i> |
|--------------|-------------------------|
| 20           | 54.47252747252747       |
| 25           | 74.9000999000999        |
| 30           | 98.55544455544455       |
| 35           | 125.4385614385614       |
| 40           | 155.5494505494505       |
| 45           | 188.8881118881119       |
| 50           | 225.4545454545454       |
| 55           | 265.2487512487512       |
| 60           | 308.2707292707293       |
| 65           | 354.5204795204795       |
| 70           | 403.998001998002        |
| 75           | 456.7032967032967       |


```


```
(%i21) res: data-predictvalues;
```

```
(%o21) [[0,-0.472527472527474],[0,0.099900099900097],[0,-0.555444555444552],[0,
0.561438561438564],[0,0.45054945054946],[0,1.111888111888106],[0,0.545454545454561],
[0,-0.248751248751205],[0,-1.270729270729248],[0,-0.520479520479512],[0,-
1.998001998001996],[0,2.296703296703299]]
```

```
(%i22) residuals: makelist([20+5*(i-1),res[i][2]],i,1,length(data));
```

```
(%o22) [[20,-0.472527472527474],[25,0.099900099900097],[30,-0.555444555444552],[35,
0.561438561438564],[40,0.45054945054946],[45,1.111888111888106],[50,
0.545454545454561],[55,-0.248751248751205],[60,-1.270729270729248],[65,-
0.520479520479512],[70,-1.998001998001996],[75,2.296703296703299]]
```

```
(%i23) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(residuals),
    xlabel="residual error");
```




(%o23)

Once again we should look at the error in relation to the size of the quantity being estimated. For all but the first data point, we can calculate the relative error as follows.

```
(%i24) percenterrors: makelist([20+5*(i-1),100*residuals[i][2]/data[i][2]],i,2,11);
```

```
(%o24) [[25,0.133200133200129],[30,-0.56678015861689],[35,0.445586159871876],[40,
0.288813750352218],[45,0.585204269414793],[50,0.241351568785204],[55,-
0.0938683957551717],[60,-0.413918329227768],[65,-0.147028113129806],[70,-
0.497015422388556]]
```

```
(%i25) wxdraw2d(  
    nticks=200,  
    color=blue,  
    point_type=6,  
    point_size=1,  
    points(percenterrors),  
    xlabel="percentage error");
```



☑ The maximum relative error of this model is about 0.6%.

☐ 3.4 Improving the Model

As indicated at the beginning of this module, a rule of thumb that is often given for safe following distance is to allow 2 seconds between your car and the car in front of you. To use this rule, you would note when the car in front of you passes some marker on or near the roadway, and then you count the time it takes you to get to the marker. This time should be at least 2 seconds. The rationale behind this rule is that if the car in front of you were suddenly to come to a complete stop, the 2-second separation distance would give you enough time to stop, to avoid hitting the vehicle in front of you. To test the 2-second rule of thumb, we will calculate how far your car travels in 2 seconds at various speeds and then compare these distances with the corresponding stopping distances in our data set.

The first entry in each element of the following table is the traveling speed in miles per hour, the second entry is the 2-second distance between the two cars in feet, and the third entry is the stopping distance in feet, taken from the test data. (To calculate the distance traveled in 2 seconds, we convert the traveling speeds from mph to ft/sec by using the conversion, 60 mph = 88 ft/sec.)

```
(%i26) data2: makelist([data[i][1],(data[i][1]*88/60.0)*2,data[i][2]],i,1,12);
```

```
(%o26) [[20,58.66666666666666,54],[25,73.33333333333333,75],[30,88.0,98],[35,102.66666666666667,126],[40,117.33333333333333,156],[45,132.0,190],[50,146.66666666666667,226],[55,161.33333333333333,265],[60,176.0,307],[65,190.66666666666667,354],[70,205.33333333333333,402],[75,220.0,459]]
```

```
(%i27) matrix(["v (mph)", "2-sec dist. (ft)", "stopping dist (ft)", [20, 58.66666666666666, 54],
[25, 73.33333333333333, 75], [30, 88.0, 98], [35, 102.66666666666667, 126],
[40, 117.33333333333333, 156], [45, 132.0, 190], [50, 146.66666666666667, 226],
[55, 161.33333333333333, 265], [60, 176.0, 307], [65, 190.66666666666667, 354],
[70, 205.33333333333333, 402], [75, 220.0, 459]);
```

```
(%o27)


| <i>v (mph)</i> | <i>2-sec dist. (ft)</i> | <i>stopping dist (ft)</i> |
|----------------|-------------------------|---------------------------|
| 20             | 58.66666666666666       | 54                        |
| 25             | 73.33333333333333       | 75                        |
| 30             | 88.0                    | 98                        |
| 35             | 102.66666666666667      | 126                       |
| 40             | 117.33333333333333      | 156                       |
| 45             | 132.0                   | 190                       |
| 50             | 146.66666666666667      | 226                       |
| 55             | 161.33333333333333      | 265                       |
| 60             | 176.0                   | 307                       |
| 65             | 190.66666666666667      | 354                       |
| 70             | 205.33333333333333      | 402                       |
| 75             | 220.0                   | 459                       |


```

The data in the table show that for speeds greater than 20 mph, the stopping distance exceeds the 2-second separation distance.

Let's find an improved model to provide the basis for a better rule of thumb. To do this, we will look at the problem the other way around. We know the stopping distance for various speeds. If we force the stopping distance and the separation distance to be equal, then we can calculate separation time for each speed. We do this by taking the stopping distance (in feet) and dividing it by the travel speed (in ft/sec). In the following table, the first entry in each element is the traveling speed in miles per hour, and the second entry is the separation time in seconds that is required to ensure a separation distance equal to the stopping distance.

```
(%i28) data3: makelist([data[i][1], data[i][2]/(data[i][1]*88/60)], i, 1, length(data));
```

```
(%o28) [[20, 1.840909090909091], [25, 2.045454545454545], [30, 2.227272727272727], [35,
2.454545454545455], [40, 2.659090909090909], [45, 2.878787878787879], [50,
3.081818181818182], [55, 3.285123966942149], [60, 3.488636363636364], [65,
3.713286713286713], [70, 3.915584415584415], [75, 4.172727272727273]]
```

```
(%i29) matrix(["speed (mph)", "separation time (sec)"], [20, 1.840909090909091],  
             [25, 2.045454545454545], [30, 2.227272727272727], [35, 2.454545454545455],  
             [40, 2.659090909090909], [45, 2.878787878787879], [50, 3.081818181818182],  
             [55, 3.285123966942149], [60, 3.488636363636364], [65, 3.713286713286713],  
             [70, 3.915584415584415], [75, 4.172727272727273]);
```


```
(%o29) 

| <i>speed (mph)</i> | <i>separation time (sec)</i> |
|--------------------|------------------------------|
| 20                 | 1.840909090909091            |
| 25                 | 2.045454545454545            |
| 30                 | 2.227272727272727            |
| 35                 | 2.454545454545455            |
| 40                 | 2.659090909090909            |
| 45                 | 2.878787878787879            |
| 50                 | 3.081818181818182            |
| 55                 | 3.285123966942149            |
| 60                 | 3.488636363636364            |
| 65                 | 3.713286713286713            |
| 70                 | 3.915584415584415            |
| 75                 | 4.172727272727273            |


```

Now we plot the separation time versus speed.

```
(%i30) wxdraw2d(  
    nticks=200,  
    color=blue,  
    point_type=6,  
    point_size=1,  
    points(data3),  
    xlabel="speed (mph)",  
    ylabel="separation time (sec)",  
    xrange=[-5,80],  
    yrange=[-1,5]);
```



It appears that the separation time is a linear function of the travel speed and not a constant function as the 2-second rule suggests. Let's find the best-fit linear function for the separation-time versus speed data and plot it together with the data points.

```
(%i31) linreg: simple_linear_regression(data3);
```

```
SIMPLE LINEAR REGRESSION
```

```
model = 0.0420523431855281 x + 0.98278349202539
```

```
correlation = 0.999793973118816
```

```
v_estimation = 2.605820886073781 10-4
```

```
(%o31) b_conf_int = [0.041450786570288 , 0.0426538998007682 ]
```

```
hypotheses = H0: b = 0 , H1: b # 0
```

```
statistic = 155.7600015071964
```

```
distribution = [student_t, 10]
```


```
p_value = 0.0
```

```
(%i32) eqn: take_inference(model,linreg);
```

```
(%o32) 0.0420523431855281 x + 0.98278349202539
```


```
(%i33) wxdraw2d(  
    nticks=200,  
    color=red,  
    explicit(eqn,x,0,85),  
    xrange=[-5,80],  
    yrange=[-1,5],  
    xlabel="speed (mph)",  
    ylabel="separation time (sec)");
```

```
(%t33)
```



```
(%o33)
```

```
(%i34) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(data3),
    color=red,
    explicit(eqn,x,0,85),
    xrange=[-5,80],
    yrange=[-1,5],
    xlabel="speed (mph)",
    ylabel="separation time (sec)");
```



3.5 You Try It: Make a Better Rules of Thumb

Using the Model

Based upon the results of the analysis in Part II, formulate a better rule of thumb for separation time versus travel speed. Keep in mind that a rule of thumb should be easy to remember, easy to use, and most importantly, it should be correct.

4 Part III: Radioactivity

4.1 Observing the Data

A radioactive dye is injected into a patient's veins to facilitate an X-ray procedure. Measuring the radioactivity in counts per minute every minute for 10 minutes yielded the table of values shown below.

```
(%i35) kill(all)$
      load(draw)$
      load(stats)$
      load(lsqares)$
      ratprint: false$
```

```
(%i5) data: [[0, 10023], [1, 8174], [2, 6693], [3, 5500], [4, 4489],
             [5, 3683], [6, 3061], [7, 2479], [8, 2045], [9, 1645], [10, 1326]];
```

```
(%o5) [[0, 10023], [1, 8174], [2, 6693], [3, 5500], [4, 4489], [5, 3683], [6, 3061], [7, 2479], [8,
2045], [9, 1645], [10, 1326]]
```

```
(%i6) datamatrix: matrix([0, 10023], [1, 8174], [2, 6693], [3, 5500], [4, 4489],
                        [5, 3683], [6, 3061], [7, 2479], [8, 2045], [9, 1645], [10, 1326]);
```

```
(%o6) [ 0  10023
      1   8174
      2   6693
      3   5500
      4   4489
      5   3683
      6   3061
      7   2479
      8   2045
      9   1645
     10   1326 ]
```

```
(%i7) matrix(["Time (min)", "Radioactivity (cpm)"], [0, 10023], [1, 8174],  
            [2, 6693], [3, 5500], [4, 4489], [5, 3683], [6, 3061], [7, 2479],  
            [8, 2045], [9, 1645], [10, 1326]);
```


```
(%o7) 

| <i>Time (min)</i> | <i>Radioactivity (cpm)</i> |
|-------------------|----------------------------|
| 0                 | 10023                      |
| 1                 | 8174                       |
| 2                 | 6693                       |
| 3                 | 5500                       |
| 4                 | 4489                       |
| 5                 | 3683                       |
| 6                 | 3061                       |
| 7                 | 2479                       |
| 8                 | 2045                       |
| 9                 | 1645                       |
| 10                | 1326                       |


```

☑ We begin by plotting the data points.

```
(%i8) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(data),
    xlabel="x (min)",
    ylabel="y (cpm)",
    xrange=[-2,10],
    yrange=[-1000,11000]);
```



4.2 Designing a Model

There appears to be a trend that we can capture with a mathematical model, and now we try to find a suitable model using the `lsquares_estimates` function. What does the function look like to you? A decaying exponential? We try to find a function of the form $y=ae^{-kx}$, where we vary the value of k and the computer finds the a to find the best-fit function of this form.

```
(%i9) numer:true$
```

```
(%i10) coeff: lsquares_estimates(datamatrix, [x,y], y=a*%e^(-0.1*x), [a]);
```

```
(%o10) [a = 7584.054545454545]
```


```
(%i11) a: rhs(coeff[1]);
```

```
(%o11) 7584.054545454545
```


```
(%i12) f2: a*%e^(-0.1*x);  
(%o12) 7584.054545454545 %e-0.1 x
```

Next, we plot our model function, save it as p2, and then show the plots of the data and the model function together on the same graph.

```
(%i13) wxdraw2d(  
    nticks=200,  
    color=red,  
    explicit(f2,x,0,12),  
    xlabel="x (min)",  
    ylabel="y (cpm)",  
    xrange=[-2,10],  
    yrange=[-1000,11000]);
```



```
(%i14) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(data),
    color=red,
    explicit(f2,x,0,12),
    xlabel="x (min)",
    ylabel="y (cpm)",
    xrange=[-2,10],
    yrange=[-1000,11000]);
```



4.3 Assessing the Errors

Our model doesn't appear to be very good! Let's quantify just how good (or bad) it is. You guessed right...we should calculate the residual errors to do this quantification.

```
(%i15) predictvalues: makelist([i,subst(i,x,f2)],i,0,10);
```

```
(%o15) [[0, 7584.054545454545],[1, 6862.336333152974],[2, 6209.29868938449],[3,
5618.40579391673],[4, 5083.74379204589],[5, 4599.961606751142],[6, 4162.217383317246],[
7, 3766.130030425165],[8, 3407.735372717596],[9, 3083.446475999599],[10,
2790.017747995555]]
```

```
(%i16) res: data-predictvalues;
```

```
(%o16) [[0,2438.945454545455],[0,1311.663666847026],[0,483.7013106155091],[0,-118.4057939167305],[0,-594.7437920458906],[0,-916.9616067511415],[0,-1101.217383317246],[0,-1287.130030425165],[0,-1362.735372717596],[0,-1438.446475999599],[0,-1464.017747995555]]
```

```
(%i17) residuals: makelist([i-1,res[i][2]],i,1,11);
```

```
(%o17) [[0,2438.945454545455],[1,1311.663666847026],[2,483.7013106155091],[3,-118.4057939167305],[4,-594.7437920458906],[5,-916.9616067511415],[6,-1101.217383317246],[7,-1287.130030425165],[8,-1362.735372717596],[9,-1438.446475999599],[10,-1464.017747995555]]
```

To get an overall measure of the error for all of the data points in the set, you might be inclined to calculate the average (mean) of all of the individual or local residuals, but this can be very misleading.

```
(%i18) residavg: sum(residuals[i][2],i,1,11)/length(residuals);
```

```
(%o18) -368.122524650994
```

As you can see, the mean residual error is small relative to the average size of the measured values of radioactivity in the data set. On this basis, we might be led to believe that our model isn't so bad after all. Wrong! The problem is that the individual or local errors are actually sizable when compared to the measured radioactivity counts, but because some of them are positive and some are negative, they tend to cancel each other when we add them together to calculate their mean value. Look at the values in the list of residual errors or look at the vertical differences between the data points and the fit function on the graph, and it is easy to see that the average of the residual errors is misleading.

There is an infinite number of ways to address this canceling problem, but one of the most common is to calculate the sum of the squares of the residuals and try to find the smallest or least value of this sum of squared residuals. Hence we have the "least squares." Squaring the residuals removes the canceling effect that occurs when we add them together for a measure of the global error. The Maxima `lsquares_estimates()` function uses least squares, and some variations of it, to find a best-fit function for a set of data. Finding the minimum value for the sum of squared residuals is a problem that can be solved using calculus, and you will study this problem later on, but for now you can do it by trial and error. Let's get back to our radioactivity problem. What we do now is to calculate the sum of the squares of the residuals for our data set.

```
(%i19) residsquaresum: sum(residuals[i][2]^2,i,1,11);  
(%o19) 1.8050349528891273 107
```

Before we compare this value, we calculate the mean of the squared residuals.

```
(%i20) msresiduals: residsquaresum/length(residuals);  
(%o20) 1640940.866262843
```

Comparing the average of the squared residuals with the average of the radioactivity counts in the data set would be like comparing apples with oranges because msresiduals is an average of squares, whereas the average radioactivity count is not an average of squared values. To make a fair comparison, we take the square root of the mean of the squares.

```
(%i21) rmsresiduals: sqrt(msresiduals);  
(%o21) 1280.99214137435
```

The value that we calculate in the preceding step is the root of the mean of the squares of the residuals and is often times called the root-mean-square or "rms" value of the residuals.

4.4 You Try It: Improving the Model and Making Predictions

IMPROVING THE MODEL

In Part III, we calculated the "rms" value of the residual errors, a number that we can use for a fair comparison with the average of the measured radioactivity values. As we initially expected, this comparison leads us to conclude that our model needs improvement. We will leave that up to you, but, to help out, we group all of the commands that you need for the error analysis into one cell, the one that follows. Use trial and error to find a better model function by changing the value of k to reduce (possibly minimize) the sum of the squared residuals (which will also minimize the "rms" value).

A comment about errors in modeling: Keep in mind that in mathematical modeling you may not be able to completely eliminate errors. This is because of random errors that occur in measurements due to the limited precision of all measuring instruments, and because of systematic errors (i.e., errors that follow a pattern) due to shortcomings of the model and/or possible defects in the measuring tools. Systematic errors can often be reduced or eliminated by refining the model and repairing or recalibrating the measuring equipment, but random errors are still unavoidable. We can reduce the magnitudes of random errors by using more precise instruments, but we cannot eliminate them. The errors may be smaller, but they are always present.

```
(%i22) kill(a);
      k: 0.2;
      coeff: lsquares_estimates(datamatrix, [x,y], y=a*%e^(-k*x), [a]);
      a: rhs(coeff[1]);
      f2: a*%e^(-k*x);
      wxdraw2d(
        nticks=200,
        color=red,
        explicit(f2,x,0,10),
        color=blue,
        point_type=6,
        point_size=1,
        points(data),
        xlabel="x (min)",
        ylabel="y (cpm)",
        xrange=[-2,10],
        yrange=[-1000,11000]);
      residuals: makelist([data[i][2],data[i][2]-subst(data[i][1],x,f2)],i,1,length(data));
      residsquaresum: sum(residuals[i][2]^2,i,1,11);
      mresiduals: residsquaresum/length(data);
      rmsresiduals: sqrt(mresiduals);
```


(%o22) done

(%o23) 0.2

(%o24) [a = 10009.890625]

(%o25) 10009.890625

(%o26) 10009.890625 %e^{-0.2 x}



(%o27)

(%o28) [[10023, 13.109375],[8174, -21.40528963448014],[6693, -16.83034456171481],[5500, 6.455548971493954],[4489, -8.733785457937302],[3683, 0.567030687740498],[3061, 46.07888187578556],[2479, 10.59136248745062],[2045, 24.03793730515531],[1645, -9.62379153218194],[1326, -28.69138290188926]]

(%o29) 4760.652840469596

(%o30) 432.7866218608724

MAKING PREDICTIONS

Now use your improved model to determine when the radioactivity will fall below 500 counts per minute.

```
(%i32) solve([f2=500], [x]);
```

```
(%o32) [x=14.98360409145608]
```

Your improved model should predict that the radioactivity will be below 500 cpm after about 15 minutes. Is that what you get?

5 Part IV: Ponderosa Pines

5.1 Observing the Data

In the table that follows, the girth of the pine tree (the distance around the tree at shoulder height) is measured in inches, and the volume of usable lumber obtained from the tree is measured in board feet (bf). We will formulate and test the following models: that usable board feet is proportional to (a) the square of the girth and (b) the cube of the girth. Which is better? Does one model provide a better explanation than the other?

```
(%i33) kill(all)$
load(draw)$
load(stats)$
load(lsqares)$
ratprint: false$
```

```
(%i5) data: [[17, 19], [19, 25], [20, 32], [23, 57], [25, 71], [28, 113], [32,123], [38, 252],
[39, 259], [41, 294]];
```

```
(%o5) [[17, 19],[19, 25],[20, 32],[23, 57],[25, 71],[28, 113],[32, 123],[38, 252],[39, 259],[
41, 294]]
```


```
(%i6) datamatrix: matrix([17, 19], [19, 25], [20, 32], [23, 57], [25, 71], [28, 113], [32,123], [38, 252],  
[39, 259], [41, 294]);
```

```
(%o6) [ 17  19  
      19  25  
      20  32  
      23  57  
      25  71  
      28 113  
      32 123  
      38 252  
      39 259  
      41 294 ]
```

```
(%i7) matrix(["girth (in)","lumber (bf)],[17, 19], [19, 25], [20, 32], [23, 57], [25, 71],  
[28, 113], [32,123], [38, 252], [39, 259], [41, 294]);
```

```
(%o7) [ girth (in)  lumber (bf)  
      17           19  
      19           25  
      20           32  
      23           57  
      25           71  
      28          113  
      32          123  
      38          252  
      39          259  
      41          294 ]
```

```
(%i8) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(data),
    xlabel="girth (in)",
    ylabel="lumber (bf)",
    xrange=[-10,50],
    yrange=[-50,300]);
```



5.2 Designing a Model

We are asked to compare a quadratic and a cubic model, so now we use the `lsquares_estimates()` function to find the best-fit function for each model.

```
(%i9) coeff1: lsquares_estimates(datamatrix, [x,y], y=a*x^2, [a]);
      a: rhs(coeff1[1]);
      y2: a*x^2;
```

```
(%o9) [a=0.157918675520786]
```

```
(%o10) 0.157918675520786
```

```
(%o11) 0.157918675520786 x^2
```

```
(%i12) coeff2: lsquares_estimates(datamatrix, [x,y], y=b*x^3, [b]);  
      b: rhs(coeff2[1]);  
      y3: b*x^3;
```


```
(%o12) [b=0.00436203466894004]
```

```
(%o13) 0.00436203466894004
```

```
(%o14) 0.00436203466894004 x3
```

```
(%i15) wxdraw2d(  
      nticks=200,  
      title="Quadratic Model",  
      color=blue,  
      explicit(y2,x,0,45),  
      xlabel="girth (in)",  
      ylabel="lumber (bf)",  
      xrange=[-10,50],  
      yrange=[-50,300]);
```


```
(%t15)
```



```
(%o15)
```


```
(%i16) wxdraw2d(  
    nticks=200,  
    title="Cubic Model",  
    color=green,  
    explicit(y3,x,0,45),  
    xlabel="girth (in)",  
    ylabel="lumber (bf)",  
    xrange=[-10,50],  
    yrange=[-50,300]);
```

```
(%t16)
```



```
(%o16)
```

```
(%i17) wxdraw2d(
    nticks=200,
    title="Combined Model",
    color=blue,
    explicit(y2,x,0,45),
    color=green,
    explicit(y3,x,0,45),
    color=black,
    point_type=6,
    point_size=1,
    points(data),
    xlabel="girth (in)",
    ylabel="lumber (bf)",
    xrange=[-10,50],
    yrange=[-50,300]);
```



The graphs seem to show that the cubic model better fits the data set.

5.3 Assessing the Errors


Now we analyze the fit for each of the models by calculating the percent errors, first for the quadratic function and then for the cubic.

ANALYSIS OF THE ERRORS FOR THE QUADRATIC-FIT FUNCTION

Let's use the model to calculate the volume of lumber for each girth measurement in the original data set, and then calculate the residual errors and plot them.

```
(%i18) quadpredictedvalues: makelist([data[i][1], subst(data[i][1],x,y2)],i,1,length(data));
      quadresiduals: makelist([data[i][1],data[i][2]-quadpredictedvalues[i][2]],i,1,length(data));
(%o18) [[17, 45.63849722550714],[19, 57.00864186300373],[20, 63.16747020831438],[23,
83.53897935049577],[25, 98.69917220049122],[28, 123.8082416082962],[32,
161.7087237332848],[38, 228.0345674520149],[39, 240.1943054671154],[41,
265.4612935504412]]
(%o19) [[17, -26.63849722550714],[19, -32.00864186300373],[20, -31.16747020831438],[23, -
26.53897935049577],[25, -27.69917220049122],[28, -10.80824160829619],[32, -
38.70872373328481],[38, 23.96543254798507],[39, 18.80569453288456],[41,
28.53870644955884]]
```

```
(%i20) wxdraw2d(
      nticks=200,
      color=red,
      point_type=6,
      point_size=1,
      points(quadresiduals),
      xlabel="girth (in)",
      ylabel="residual error");
```




Note that the errors aren't completely random. There are more negative errors than there are positive ones, and the error seems to increase as the girth increases. These observations suggest that the quadratic function is not appropriate for modeling the relation between the volume of usable lumber in a tree and its girth.

It is usually more helpful to look at the error in relation to the size of the quantity being estimated. We calculate the relative percent errors as follows.

```
(%i21) quadpercenterrors: makelist([data[i][1],
    100*(data[i][2]-quadpredictedvalues[i][2])/data[i][2],i,1,length(data)];
(%o21) [[17,-140.2026169763533],[19,-128.0345674520149],[20,-97.39834440098242],[23,-
46.5596128956066],[25,-39.01291859224116],[28,-9.564815582562995],[32,-
31.47050710023156],[38,9.510092280946457],[39,7.260885920032648],[41,
9.70704301005403]]
```

Now we plot the percent errors.

```
(%i22) wxdraw2d(
    nticks=200,
    color=red,
    point_type=6,
    point_size=1,
    points(quadpercenterrors),
    xrange=[-5,45],
    yrange=[-150,20],
    xlabel="girth (in)",
    ylabel="percent error");
```




The percent error of largest magnitude is about -140%, and there is a trend in the errors, indicating that they are not random.

ANALYSIS OF THE ERRORS FOR THE CUBIC-FIT FUNCTION

Let's look at the errors for the cubic model. We do the same as we did for the quadratic function.

```
(%i23) cubicpredictedvalues: makelist([data[i][1], subst(data[i][1],x,y3)],i,1,length(data));
      cubicresiduals: makelist([data[i][1],data[i][2]-cubicpredictedvalues[i][2]],i,1,length(data));
(%o23) [[17,21.43067632850241],[19,29.91919579425973],[20,34.89627735152032],[23,
53.07287581699346],[25,68.15679170218812],[28,95.75538505257175],[32,
142.9351520318272],[38,239.3535663540779],[39,258.7515345268542],[41,
300.6357914180165]]
(%o24) [[17,-2.430676328502415],[19,-4.919195794259732],[20,-2.896277351520318],[23,
3.92712418300654],[25,2.843208297811884],[28,17.24461494742825],[32,-
19.93515203182722],[38,12.64643364592214],[39,0.248465473145814],[41,-
6.635791418016481]]
```


```
(%i25) wxdraw2d(
      nticks=200,
      color=blue,
      point_type=6,
      point_size=1,
      points(cubicresiduals),
      xlabel="girth (in)",
      ylabel="residual error");
```



Note that the residuals exhibit a more random pattern than they did in the quadratic model. There are as many negative errors as positive ones, and there is no apparent trend in the errors.

```
(%i26) cubicpercenterrors: makelist([data[i][1],
    100*(data[i][2]-cubicpredictedvalues[i][2])/data[i][2]],i,1,length(data));
(%o26) [[17,-12.79303330790744],[19,-19.67678317703893],[20,-9.050866723500995],[23,
6.889691549134282],[25,4.004518729312513],[28,15.26072119241438],[32,-
16.20744067628229],[38,5.018426049969104],[39,0.0959326151142137],[41,-
2.25707191088996]]
```

```
(%i27) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(cubicpercenterrors),
    xlabel="girth (in)",
    ylabel="percent error");
```




(%o27)

In this case, the percent error of largest magnitude is about -20%, which is much better than for the quadratic regression function.

COMPARE THE ERRORS

To confirm our earlier assessment that the cubic regression is evidently better, we now plot the percent errors for the two models together.

```
(%i28) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(cubicpercenterrors),
    color=red,
    points(quadpercenterrors),
    xlabel="girth (in)",
    ylabel="percent error");
```



5.4 Explaining the Model

The unit of board feet is a measure of volume. If a tree is modeled as a right circular cone, its volume is approximated by $V = (\pi r^2 h)/3$ where V is the volume, h is the height of the tree, and r is its radius. The girth, g , is the circumference of the tree near the base so that $g = 2\pi r$ or $r = g/(2\pi)$. If we assume that as a tree grows the proportion $h/r = k$ is a constant, then we have that

$$V = (\pi/3) * (g/(2\pi))^2 * (kg/(2\pi)) = k/(24\pi^2) * g^3,$$

where the last quantity $k/(24\pi^2)$, is a constant. This shows that cubic relation between the volume of lumber produced and the girth of the tree near its base is a rational model.

6 Part V: Black Bass

6.1 Observing the Data

In the table that follows, L represents the lengths of New York black bass measured in inches (in.), and w represents the weight of the fish. Formulate and test a model that assumes the weight of the fish is proportional to the cube of its length. Can you provide a rational explanation of the model?

```
(%i29) kill(all)$
      load(draw)$
      load(stats)$
      load(lsqares)$
      ratprint: false$
```

```
(%i5) data: [[12.5, 17], [12.63, 16], [14.13, 17], [14.5, 23], [14.5, 26], [14.5, 27],
             [17.25, 41], [17.75, 49]];
```

```
(%o5) [[12.5, 17], [12.63, 16], [14.13, 17], [14.5, 23], [14.5, 26], [14.5, 27], [17.25, 41], [17.75, 49]
]]
```

```
(%i6) datamatrix: matrix([12.5, 17], [12.63, 16], [14.13, 17], [14.5, 23], [14.5, 26], [14.5, 27],
                          [17.25, 41], [17.75, 49]);
```


```
(%o6)
      12.5  17
      12.63 16
      14.13 17
      14.5  23
      14.5  26
      14.5  27
      17.25 41
      17.75 49
```

```
(%i7) matrix(["L (in)", "W (oz)"], [12.5, 17], [12.63, 16], [14.13, 17], [14.5, 23], [14.5, 26],
             [14.5, 27], [17.25, 41], [17.75, 49]);
```

```
(%o7)
      L (in)  W (oz)
      12.5    17
      12.63   16
      14.13   17
      14.5    23
      14.5    26
      14.5    27
      17.25   41
      17.75   49
```

Next, we plot the data to see if there is a recognizable pattern.

```
(%i8) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(data),
    xlabel="L (in)",
    ylabel="w (oz)",
    xrange=[0,20],
    yrange=[-5,50]);
```



6.2 Designing a Model

We are asked to construct a cubic model, so now we find the best-fit function using the `lsquares_estimates()` command.

```
(%i9) coeff: lsquares_estimates(datamatrix, [x,y], y=a*x^3, [a]);
a: rhs(coeff[1]);
y4: a*x^3;
```


```
(%o9) [a=0.00818627962827608]
```

```
(%o10) 0.00818627962827608
```

```
(%o11) 0.00818627962827608 x3
```

Next, we plot the model function and superimpose its graph on the graph of the data.

```
(%i12) wxdraw2d(  
    nticks=200,  
    color=blue,  
    explicit(y4,x,0,45),  
    color=black,  
    point_type=6,  
    point_size=1,  
    points(data),  
    xlabel="L (in)",  
    ylabel="w (oz)",  
    xrange=[0,20],  
    yrange=[-5,50]);
```




✓ The graphs show that the cubic model fits the data well.

□ 6.3 Assessing the Errors

✓ Now we analyze the fit by calculating the percent errors. First, we use the model to calculate the weight of the fish for each length measurement in the original data set, and then we calculate the residuals and plot them.

```
(%i13) cubicpredictedvalues: makelist([data[i][1], subst(data[i][1],x,y4)],i,1,length(data));
      cubicresiduals: makelist([data[i][1],data[i][2]-cubicpredictedvalues[i][2]],i,1,length(data));
(%o13) [[12.5, 15.98882739897672],[12.63, 16.49288485379556],[14.13, 23.09473912131148],[
14.5, 24.95689673175316],[14.5, 24.95689673175316],[14.5, 24.95689673175316],[17.25,
42.01978960008353],[17.75, 45.78061762556124]]
(%o14) [[12.5, 1.011172601023285],[12.63, -0.492884853795555],[14.13, -6.094739121311484]
,[14.5, -1.956896731753158],[14.5, 1.043103268246842],[14.5, 2.043103268246842],[17.25, -
1.019789600083534],[17.75, 3.219382374438759]]
```

```
(%i15) wxdraw2d(
      nticks=200,
      color=blue,
      point_type=6,
      point_size=1,
      points(cubicresiduals),
      xlabel="L (in)",
      ylabel="residual error");
```




(%o15)

Note that there are nearly equal numbers of positive and negative errors, and there is no apparent trend in the errors. They appear to be random.

As before, it is more meaningful to look at the error in relation to the size of the quantity being estimated, so we calculate the relative percent errors and plot them.

```
(%i16) cubicpercenterrors: makelist([data[i][1],
    100*(data[i][2]-cubicpredictedvalues[i][2])/data[i][2]],i,1,length(data));
(%o16) [[12.5, 5.94807412366638],[12.63, -3.08053033622222],[14.13, -35.8514065959499],[
14.5, -8.50824665979634],[14.5, 4.011935647103238],[14.5, 7.56704914165497],[17.25, -
2.487291707520815],[17.75, 6.570168111099508]]
```

```
(%i17) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(cubicpercenterrors),
    xlabel="L (in)",
    ylabel="percent error");
```



6.4 Improving the Model


The largest percent error about -36%, but this data point seems to be an outlier. Let's see what happens if we remove this point from the data set.

```
(%i18) data: [[12.5, 17], [12.63, 16], [14.5, 23], [14.5, 26], [14.5, 27],
    [17.25, 41], [17.75, 49]];
(%o18) [[12.5, 17],[12.63, 16],[14.5, 23],[14.5, 26],[14.5, 27],[17.25, 41],[17.75, 49]]
```

```
(%i19) datamatrix: matrix([12.5, 17], [12.63, 16], [14.5, 23], [14.5, 26], [14.5,27],  
[17.25, 41], [17.75, 49]);
```

```
(%o19) [ 12.5  17  
12.63  16  
14.5   23  
14.5   26  
14.5   27  
17.25  41  
17.75  49]
```

```
(%i20) wxdraw2d(  
nticks=200,  
color=blue,  
point_type=6,  
point_size=1,  
points(data),  
xrange=[0,20],  
yrange=[0,50],  
xlabel="L (in)",  
ylabel="w (oz)");
```



(%o20)

```
(%i21) kill(coeff,a);
```

```
(%o21) done
```

```
(%i22) coeff: lsquares_estimates(datamatrix, [x,y], y=a*x^3, [a]);
      a: rhs(coeff[1]);
      y5: a*x^3;
```


```
(%o22) [a=0.0083704153843138]
```

```
(%o23) 0.0083704153843138
```

```
(%o24) 0.0083704153843138 x3
```

```
(%i25) wxdraw2d(
      nticks=200,
      color=blue,
      explicit(y5,x,0,45),
      color=black,
      point_type=6,
      point_size=1,
      points(data),
      xlabel="L (in)",
      ylabel="w (oz)",
      xrange=[0,20],
      yrange=[-5,50]);
```

```
(%t25)
```



```
(%o25)
```


```
(%i26) cubicpredictedvalues: makelist([data[i][1], subst(data[i][1],x,y5)],i,1,length(data));
      cubicresiduals: makelist([data[i][1],data[i][2]-cubicpredictedvalues[i][2]],i,1,length(data));
```

```
(%o26) [[12.5, 16.3484675474879],[12.63, 16.86386287552193],[14.5, 25.51825760100366],[
14.5, 25.51825760100366],[14.5, 25.51825760100366],[17.25, 42.9649498044616],[17.75,
46.81037094711152]]
```

```
(%o27) [[12.5, 0.651532452512104],[12.63, -0.863862875521928],[14.5, -2.518257601003665],[
14.5, 0.481742398996335],[14.5, 1.481742398996335],[17.25, -1.964949804461604],[17.75,
2.189629052888478]]
```

```
(%i28) cubicpercenterrors: makelist([data[i][1],
    100*(data[i][2]-cubicpredictedvalues[i][2])/data[i][2]],i,1,length(data));
(%o28) [[12.5, 3.832543838306493],[12.63, -5.39914297201205],[14.5, -10.94894609132028],[
14.5, 1.852855380755135],[14.5, 5.487934811097538],[17.25, -4.792560498686838],[17.75,
4.468630720180567]]
```

```
(%i29) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(cubicpercenterrors),
    xlabel="L (in)",
    ylabel="percent error");
```



After deleting the one data point, we find that the new model gives a largest percent error of about -11%, which is considerably better.

6.5 Explaining the Model

The weight of a fish is proportional to its volume. If we think of a fish to be shaped roughly like an ellipsoid, then its volume would be approximated by $V = (4\pi L w d)/3$ where L , w , and d are its length, width, and depth, respectively. If, as the fish grows, we assume that its proportions remain the same then $w = k(w)L$ and $d = k(d)L$, where $k(w)$ and $k(d)$ are constants. Therefore,

$$V = (4\pi L (k(w)L) (k(d)L))/3 = (4\pi k(w)k(d)/3)L^3$$

where $(4\pi k(w)k(d)/3)$ is a constant. Therefore, the weight is proportional to the volume, which is proportional to the length cubed, and our model has a rational basis.

7 Part VI: Heart Rates of Mammals

7.1 Observing the Data

The following data relate the weight in grams (g) of some mammals to their heart rate in beats per minute. Plot the data. Is there a trend? If so, find a function that captures the trend of the data. Hint: try the form $y = x^{(-1/n)}$ for n an integer.

```
(%i30) kill(all)$
      load(draw)$
      load(stats)$
      load(lsqares)$
      ratprint: false$
```

```
(%i5) data: [[4, 660], [25, 670], [200, 420], [300, 300], [2000, 205],
             [5000,120], [30000, 85], [50000, 70], [70000, 72], [450000, 38],
             [500000,40], [3000000, 48]];

(%o5) [[4,660],[25,670],[200,420],[300,300],[2000,205],[5000,120],[30000,85],[50000,
70],[70000,72],[450000,38],[500000,40],[3000000,48]]
```

```
(%i6) datamatrix: matrix([4, 660], [25, 670], [200, 420], [300, 300], [2000, 205],
    [5000,120], [30000, 85], [50000, 70], [70000, 72], [450000, 38],
    [500000,40], [3000000, 48]);
```


```
(%o6) [ 4 660
      25 670
      200 420
      300 300
      2000 205
      5000 120
      30000 85
      50000 70
      70000 72
      450000 38
      500000 40
      3000000 48]
```

```
(%i7) matrix(["x (g)", "y (bpm)"],[4, 660], [25, 670], [200, 420], [300, 300], [2000, 205],
    [5000,120], [30000, 85], [50000, 70], [70000, 72], [450000, 38],
    [500000,40], [3000000, 48]);
```

```
(%o7) [ x (g)  y (bpm)
      4 660
      25 670
      200 420
      300 300
      2000 205
      5000 120
      30000 85
      50000 70
      70000 72
      450000 38
      500000 40
      3000000 48]
```

Next, we plot the data to see if there is a recognizable pattern.

```
(%i8) wxdraw2d(
    nticks=200,
    color=blue,
    point_type=6,
    point_size=1,
    points(data),
    xlabel="x (g)",
    ylabel="y (bpm)",
    xrange=[-5000,70000],
    yrange=[-50,700]);
```



Note that we did not plot the data points for the horse, the ox, and the elephant because that makes the scale too large to see if there is a pattern for the smaller mammals; however, these data points are included in the calculations that follow.

7.2 Designing a Model

The graph of the data suggests that there is a relationship. As suggested in the hint above, we now build a group of functions of the form $y = x^{-1/n}$ where n is an integer. For integer values of n (1, 2, 3, 4, 5), we use the `lsquares_estimates()` function to find the function of the form $x^{-1/n}$ that best fits the data.

```
(%i9) for n: 1 thru 5 do block(  
    kill(a),  
    coeff: lsquares_estimates(datamatrix, [x,y], y=a*x^(-1.0/n), [a]),  
    a: rhs(coeff[1]),  
    y[n]: a*x^(-1.0/n));
```

```
(%o9) done
```

```
(%i10) y[1];  
y[2];  
y[3];  
y[4];  
y[5];
```

```
(%o10)  $\frac{3040.893081761006}{x^{1.0}}$ 
```


```
(%o11)  $\frac{1733.5583333333333}{x^{0.5}}$ 
```

```
(%o12)  $\frac{1371.6083333333333}{x^{0.3333333333333333}}$ 
```

```
(%o13)  $\frac{1149.868312757202}{x^{0.25}}$ 
```


```
(%o14)  $\frac{989.4577777777778}{x^{0.2}}$ 
```

```
(%i15) wxdraw2d(  
    nticks=200,  
    color=red,  
    explicit(y[1],x,0,70000),  
    color=green,  
    explicit(y[2],x,0,70000),  
    color=blue,  
    explicit(y[3],x,0,70000),  
    color=black,  
    explicit(y[4],x,0,70000),  
    color=brown,  
    explicit(y[5],x,0,70000),  
    xlabel="x (g)",  
    ylabel="y (bpm)",  
    xrange=[-5000,70000],  
    yrange=[-50,700]);
```



Now we plot the models together with the data points.

```
(%i16) wxdraw2d(
    nticks=200,
    color=red,
    explicit(y[1],x,0,70000),
    color=green,
    explicit(y[2],x,0,70000),
    color=blue,
    explicit(y[3],x,0,70000),
    color=black,
    explicit(y[4],x,0,70000),
    color=brown,
    explicit(y[5],x,0,70000),
    color=purple,
    point_type=6,
    point_size=1,
    points(data),
    xlabel="x (g)",
    ylabel="y (bpm)",
    xrange=[-5000,70000],
    yrange=[-50,700]);
```




7.3 Assessing the Errors

It appears that `y[4]` provides the best fit. Now we will analyze the errors. First, we calculate the heart rate values that our selected model predicts for each mammal, and then we calculate the residual errors and plot them.

```
(%i17) predictedvalues: makelist([data[i][1], subst(data[i][1],x,y[4])],i,1,length(data));
residuals: makelist([i,data[i][2]-predictedvalues[i][2]],i,1,length(data));
wxdraw2d(
  nticks=200,
  color=blue,
  point_type=6,
  point_size=1,
  points(residuals),
  xrange=[2,12],
  yrange=[-175,175],
  xlabel="x (g)",
  ylabel="residuals");
percenterrors: makelist([i,100*(data[i][2]-predictedvalues[i][2])/data[i][2]],i,1,length(data));
wxdraw2d(
  nticks=200,
  color=blue,
  point_type=6,
  point_size=1,
  points(percenterrors),
  xrange=[2,12],
  yrange=[-30,50],
  xlabel="x (g)",
  ylabel="percent error");
```

```
(%o17) [[4,813.0796814221512],[25,514.2367424996183],[200,305.7669964881841],[300,
276.2916706644876],[2000,171.9454180046383],[5000,136.7431578847038],[30000,
87.37109778329082],[50000,76.8963286155975],[70000,70.6925416137187],[450000,
44.39611602590912],[500000,43.24198333596766],[3000000,27.62916706644875]]
```

```
(%o18) [[1,-153.0796814221512],[2,155.7632575003817],[3,114.2330035118159],[4,
23.70832933551242],[5,33.0545819953617],[6,-16.74315788470381],[7,-2.371097783290821],
[8,-6.896328615597497],[9,1.307458386281297],[10,-6.396116025909123],[11,-
3.24198333596766],[12,20.37083293355125]]
```



(%o19)

```
(%o20) [[1,-23.19389112456837],[2,23.24824738811667],[3,27.19833416947997],[4,
```

The errors appear to be somewhat random, but there is possibly a trend in the errors, with two outliers. Can you see the trend and the outliers? The largest relative percentage errors are for the largest and the smallest animals (i.e., the bat and the elephant) with magnitudes of 23% and 42%, respectively. The model appears to capture a trend in the data, which could be useful in understanding the relationship between mammal size and heart rate; however, it probably would not be useful as a predictive tool since the magnitudes of the individual errors are so large.

7.4 You Try It: A Rational Explanation for Heart Rate

Try to provide a rational explanation for the heart rate model that we created in Part VI.

8 You Try It: Modeling Takes Practice

Mathematical modeling and a computer algebra system like Maxima help us describe real-world phenomena, understand the mechanisms behind the behavior, and make predictions. The procedure for modeling generally includes the following steps:\

1. Plot and observe the data, looking for relationships and patterns.
2. Formulate a function to model the data.
3. Assess your model by analyzing the residual errors and/or relative errors.
4. Improve your model, if possible.
5. Use your model to gain a better understanding of the phenomenon you are modeling.
6. Use your model to make predictions.

With some practice, you too can become an effective mathematical modeler. There are plenty of modeling exercises that you can do yourself. Select a problem and obtain some data (possibly by designing your own experiment) and use the modeling procedures outlined in the preceding Parts of this module as a template for mathematical modeling. To help you get started, we include the data sets for four modeling examples.

8.1 Drug Levels

```
(%i22) drugdata: [[0, 853], [1, 587], [2, 390], [3, 274], [4, 189], [5, 130],
                 [6, 97], [7, 67], [8, 50], [9, 40], [10, 31]];
```

```
(%o22) [[0, 853], [1, 587], [2, 390], [3, 274], [4, 189], [5, 130], [6, 97], [7, 67], [8, 50], [9, 40], [10, 31]]
```

□ 8.2 Cell Count

```
(%i23) cellldata: [[0, 597], [2, 893], [4, 1339], [6, 1995], [8, 2976], [10,
    4433],
    [12, 6612], [14, 9865], [16, 14719], [18, 21956], [20, 32763]];
(%o23) [[0, 597], [2, 893], [4, 1339], [6, 1995], [8, 2976], [10, 4433], [12, 6612], [14, 9865], [16,
14719], [18, 21956], [20, 32763]]
```

□ 8.3 Vacuum Pump

```
(%i24) vacuumdata: [[0, 100000], [1, 36788], [2, 13537], [3, 4986], [4, 1837],
    [5, 671]];
(%o24) [[0, 100000], [1, 36788], [2, 13537], [3, 4986], [4, 1837], [5, 671]]
```

□ 8.4 Doctoral Degrees

```
(%i25) doctoraldata: [[6, 520], [10, 460], [14, 680], [18, 630], [20, 730],
    [21, 810], [22, 830]];
(%o25) [[6, 520], [10, 460], [14, 680], [18, 630], [20, 730], [21, 810], [22, 830]]
```